Занятие 5 «Методы оптимизации»

а) Вычисление производных.

Для вычисления производных Maple имеет следующие основные функции:

diff(a,x1,x2,…xn) — **diff(a,[x1,x2,…xn])**

Diff(a,x1,x2,…xn) — **Diff(a,[x1,x2,…xn])**

Здесь a-дифференцируемое алгебраическое выражение, в частности функция f(x1,x2,…,xn) переменных, по которым проводится дифференцирование.

diff(f(x),x,y) эквивалентно diff(diff(f(x),x,y).

Опреатор $ можно использовать для вычисления производных высокого порядка.

diff(f(x),x$4) вычисляет производную 4-го порядка.

Примеры:

```
> restart;
> Diff(2*x^5,x)=diff(2*x^5,x);
```

$$\frac{d}{dx}(2\,x^5) = 10\,x^4$$

```
> f(x,y):=cos(x)*y^3;
```

$$\mathrm{f}(x, y) := \cos(x)\,y^3$$

```
> Diff(f(x,y),x)=diff(f(x,y),x);
```

$$\frac{\partial}{\partial x}(\cos(x)\,y^3) = -\sin(x)\,y^3$$

```
> Diff(f(x,y),y)=diff(f(x,y),y);
```

$$\frac{\partial}{\partial y}(\cos(x)\,y^3) = 3\cos(x)\,y^2$$

```
> Diff(f(x,y),x,y)=diff(f(x,y),x,y);
```

$$\frac{\partial^2}{\partial y\,\partial x}(\cos(x)\,y^3) = -3\sin(x)\,y^2$$

```
> Diff(f(x,y),x$4)=diff(f(x,y),x$4);
```

$$\frac{\partial^4}{\partial x^4}(\cos(x)\,y^3) = \cos(x)\,y^3$$

Для вычисления производных может также использоватся дифференциальный оператор D(f) или D[i](f), где f-выражение или функция, i-положительное целое число, выражение или последовательность.

```
> f:=(x,y,z)->x*exp(y)+ln(z);
```

$$f := (x, y, z) \rightarrow x\,\mathbf{e}^y + \ln(z)$$

```
> D[1](f);
```

$$(x, y, z) \rightarrow \mathbf{e}^y$$

```
> D[2](f);
```

$$(x, y, z) \rightarrow x\,\mathbf{e}^y$$

```
> D[3](f);
```

$$(x, y, z) \rightarrow \frac{1}{z}$$

Для неявно заданной функции

```
> restart;
> f1:=x*y^2=y;
```

$$f1 := x\,y^2 = y$$

```
> implicitdiff(f1,y,x);
```

$$-\frac{y^2}{2\,x\,y-1}$$

б) Поиск экстремумов функций по нулям первой производной.

Для простых функций одной переменной f(x) поиск экстремумов сводится к нахождению точек, в которых первая производная обращается в нуль.

Примеры:

```
> restart;
> with(plots):
> y1:=expand((x-3)*(x-1)*x*(x+2));
```

$$y1 := x^4 - 2\,x^3 - 5\,x^2 + 6\,x$$

```
> dy1:=simplify(diff(y1,x));
```

$$dy1 := 4\,x^3 - 6\,x^2 - 10\,x + 6$$

```
> q1:=plot(y1,x=-5..5,y=-10..10,color=red,thickness=[3]):
> q2:=plot(dy1,x=-5..5,y=-10..10,color=black,thickness=[3]):
> display(q1,q2);
```




```
>

> extrem:=fsolve(dy1=0,x);
```

$$extrem := -1.302775638\,,\,0.5000000000\,,\,2.302775638$$

```
> extrem:=evalf(solve(dy1=0,x));
```

$$extrem := 0.5000000000\,,\,2.302775638\,,\,-1.302775638$$

в) Поиск экстремумов в аналитическом виде.

Пример:

```
> restart;
> y:=exp(-a*x)-exp(-b*x);
```

$$y := \mathbf{e}^{(-a\,x)} - \mathbf{e}^{(-b\,x)}$$

```
> dy:=diff(y,x);
```

$$dy := -a\, \mathbf{e}^{(-a\,x)} + b\, \mathbf{e}^{(-b\,x)}$$

```
> solve(dy,x);
```

$$\frac{\ln\left(\frac{a}{b}\right)}{a-b}$$

```
> restart;
```

<u>Для функции двух переменных</u>

```
> z:=(x,y)->a*x^2+b*x*y+c*y^2+d*(x-y);
```

$$z := (x, y) \rightarrow a\,x^2 + b\,x\,y + c\,y^2 + d\,(x-y)$$

```
> xy:=solve({diff(z(x,y),x)=0,diff(z(x,y),y)=0},{x,y});
```

$$xy := \{x = -\frac{d\,(b+2\,c)}{-b^2+4\,c\,a},\, y = \frac{d\,(2\,a+b)}{-b^2+4\,c\,a}\}$$

```
> z(rhs(xy[1]),rhs(xy[2]));
```

$$\frac{a\,d^2\,(b+2\,c)^2}{(-b^2+4\,c\,a)^2} - \frac{b\,d^2\,(b+2\,c)\,(2\,a+b)}{(-b^2+4\,c\,a)^2} + \frac{c\,d^2\,(2\,a+b)^2}{(-b^2+4\,c\,a)^2} + d\left(-\frac{d\,(b+2\,c)}{-b^2+4\,c\,a} - \frac{d\,(2\,a+b)}{-b^2+4\,c\,a}\right)$$

г) Поиск экстремумов с помощью функции extrema.

extrema(expr,constrs,vars,’s’)

constrs-ограничения, vars-переменные, найденные координаты точки экстремума присваиваются переменной ’s’.

При отсутствии ограничений в виде равенств или неравенств вместо них записывается пустой список {}.

Функция extrema дает неплохие результаты **при поиске экстремумов простых функций. При анализе сложных функций она часто отказывается работать и просто повторяет запись обращения к ней.**

Пример:

```
> restart;
> z:=(x,y)->a*x^2+b*x*y+c*y^2+d*(x-y);
```

$$z := (x, y) \rightarrow a\,x^2 + b\,x\,y + c\,y^2 + d\,(x-y)$$

```
> extrema(z(x,y),{},{x,y},'s');
```

$$\{-\frac{d^2\,(b+c+a)}{-b^2+4\,c\,a}\}$$

```
> s;
```

$$\{\{x = -\frac{d\,(b+2\,c)}{-b^2+4\,c\,a},\, y = \frac{d\,(2\,a+b)}{-b^2+4\,c\,a}\}\}$$

Пример2

```
> restart;
> extrema( a*x^2+b*x+c,{},x,'s');
```

$$\{\frac{-b^2+4\,c\,a}{4\,a}\}$$

```
> s;
```

$$\{\{x = -\frac{b}{2\,a}\}\}$$

д) Поиск минимумов и максимумов аналитических функций.

minimize(expr,opt1,opt2,…optn)

maximize(expr,opt1,opt2,…optn)

opt1,opt2,…optn-дополнительные данные для поиска (‘infiunity’-поиск минимума или максимума выполняется по всей числовой прямой, ‘location’-выдает не только значение минимума, но и значения переменных в точке).

```
> restart;
> minimize(x^2-3*x+y^2+3*y+3);
```

$$\frac{-3}{2}$$

```
> minimize(x^2-3*x+y^2+3*y+3,location);
```

$$\frac{-3}{2}, \left\{\left[\left\{x=\frac{3}{2}, y=\frac{-3}{2}\right\}, \frac{-3}{2}\right]\right\}$$

```
> restart;
> with(plots):
> fn := x^2-3*x+y^2+3*y+3:
> plot3d(fn, x=-10..10, y=-40..40);
```



е) Линейное программирование.

Используемые функции *minimize*() и *maximize*()

Задача линейного программирования заключается в определении наибольшего или наименьшего значений линейной функции при наличии *линейных ограничений*.

Задачи линейного программирования решаются встроенными функциями *mi nimize*() и *maximize*(), входящими в пакет *simplex*, который вызывается обычнымобразом:

> *with*(*simplex*):

Вызов пакета обязателен, т.к. входящие в ядро системы Maple 13 встроенны функции *minimize*() и *maximize*(), отличаются от рассматриваемых наборами параметров.

Структура у функций *minimize*() и *maximize*() следующая:

- *minimize*(целевая функция, {ограничения}, NONNEGATIVE);
- *maximize*(целевая функция, {ограничения}, NONNEGATIVE).

Здесь параметр NONNEGATIVE показывает, что входящие переменные неотрицательны, следовательно, в такой конструкции включать условия неотрицтельности переменных в ограничения не требуется.

Функция *feasible*({ограничения}, NONNEGATIVE), содержащаяся в пакет *simplex*, предназначена для выяснения вопроса о существовании решения для данной системы ограничений.

Например,

```
> restart;
> with(simplex):
> feasible({4*x+3*y<=5,3*x+4*y=4},NONNEGATIVE);
```

Пример:

Завод производит два вида продукции: велосипеды и мотоциклы. При этом цех по сборке велосипедов имеет мощность 100 тыс. штук в год, цех по сборке мотоциклов-30 тыс. Механические цеха завода оснащены взаимозаменяемым оборудованием, и одна группа цехов может производить либо детали для 120тыс.велосипедов, либо детали для 40 тыс. мотоциклов, либо любую другую комбинацию деталей, ограниченную этими данными. Другая группа механических цехов может выпускать детали либо для 80 тыс. велосипедов, либо для 60 тыс. мотоциклов, либо любую допустимую их комбинацию. В результате реализации каждой тысячи велосипедов завод получит прибыль в 2 тыс. у.е., а каждой тысячи мотоциклов-3тыс.у.е.

Найти такое сочетание выпусков продукции, которое дает наибольшую сумму прибыли.

<u>Решение</u>

Обозначим через x_1 и x_2 соответственно количества велосипедов и мотоциклов, выпускаемые заводом в год (в тыс. шт.)

Учитывая возможности сборочных цехов, необходимо потребовать

$$x_1 \le 100, \quad x_2 \le 30.$$

Если предусматривается производство 1000 велосипедов, то доля занятой производственной мощности механических цехов первой группы составит 1/120 всей их мощности, принимаемой в данном случае за единицу. На выпуск же x_1 тыс. велосипедов потребуется занять *1/120*x_1 всей мощности $\begin{pmatrix}120- & 1(100\%)\\ x_1- & x\end{pmatrix}$. Аналогично для производства x_2 тыс. мотоциклов необходимо выделить *1/40*x_2 всей мощности. Но в производственном процессе может быть использовано не более чем вся наличная производственная мощность, следовательно

$$1/120x_1 + 1/40x_2 \le 1.$$

Аналогично для цехов второй группы

$$1/80x_1 + 1/60x_2 \le 1.$$

По смыслу задачи $x_1 \ge 0, x_2 \ge 0$.

Любой план (x_1, x_2), удовлетворяющий данным ограничениям, будет допустимым и даст прибыль (в тыс.у.е.)

$$f = 2 * x_1 + 3 * x_2$$

Итак, математическая модель данной задачи: найти значения x_1^* и x_2^* при которых линейная функция f достигает максимального значения при ограничениях

$$\begin{cases} x_1 \le 100 \\ x_2 \le 30 \\ 1/120x_1 + 1/40x_2 \le 1 \\ 1/80x_1 + 1/60x_2 \le 1 \\ x_1 \ge 0, x_2 \ge 0 \end{cases}$$

```
> restart;
> with(simplex):
> f:=2*x1+3*x2;
```

$$f := 2\,x1 + 3\,x2$$

```
>
maximize(f,{x1<=100,x2<=30,1/120*x1+1/40*x2<=1,1/80*x1+1/60*x2<=1},NONNEGATIVE);
```

$$\{x1 = 48, x2 = 24\}$$

```
> subs(%,f);
```

$$168$$

Итак, выпускать следует 48 тыс. велосипедов и 24 тыс. мотоциклов; максимальная прибыль завода по этим видам продукции составит 168 тыс. у.е.

ж) Нелинейное программирование.

Нелинейное программирование позволяет решать задачи оптимизации при нелинейных зависимостях целевой функции от ее аргументов.

```
> restart;
> with(Optimization):
> NLPSolve(w^3*(v-w)^2+(w-x-1)^2+(x-y-2)^2+(y-z-3)^2,
{w+x+y+z<=5,  3*z+2*v-3=0}, assume = nonnegative);
```

$$[6.43845963876504790, [v = 1.50000000000000000, w = 1.68857375382855278, x = 1.93714208200857940, y = 1.37428416416286824, z = 0.]]$$

Задания

1. Максимизируйте линейную функцию $z = 2x_1 + 2x_2$ при ограничениях $x_1 \le 3,\ \ 3x_1 - 2x_2 \ge -6,\ \ 3x_1 + x_2 \ge 3.$
2. Максимизируйте линейную функцию $z = x_1 + 3x_2 + 3x_3$ при ограничениях $x_2 + x_3 \le 3,\ \ x_1 - x_2 \ge 0,\ \ x_2 \ge 1,\ \ 3x_1 + x_2 \le 15.$

3. Найдите решения, минимизирующие линейную форму $z = 12x_1 + 5x_2 + 3x_3$, на множестве неотрицательных решений системы уравнений
   $4x_1 + 3x_2 + x3 = 180, \quad 4x_2 + 9x_3 + 12x_4 = 900$.
4. Максимизируйте линейную функцию $z = 2x_1 + 3x_2 - 6x3 + 8x4 + 10$ на множестве неотрицательных значений функции при ограничениях
   $2x_1 + 3x_2 - x_3 + x_4 = 1, \quad 8x_1 + 12x_2 - 9x_3 + 8x_4 = 3, \quad 4x_1 + 6x_2 + 3x_3 - 2x_4 = 3$
5. Используя функцию нелинейного программирования NLPSolve решить задачу: пусть требуется спроектировать контейнер в форме прямоугольного параллелепипеда объемом **1**$м^3$. Израсходовать на его изготовление как можно меньше материала. Пусть данный контейнер имеет длину не менее 2м.
6. Вычислить первую производную функции $y = e^{2x - x^2}$. Найти экстремумы с помощью функции extrema и построить график функции и ее производной.
7. Найти значение первой и второй производной функции $y = x\cos 2x / \sin x$
8. Вычислить y''' для функции $y = (x+1)\ln(x+1)$.
9. Найти вторые частные производные функции $z = \cos(x^2 y^2 - 5)$.